

## SYMBOLIC MOMENT CALCULATION FOR THE SOJOURN TIME IN M/G/1 QUEUES WITH BERNOULLI FEEDBACK

Hideaki Takagi      Ken-ichi Sakamaki  
*University of Tsukuba*

(Received January 7, 1998)

*Abstract* In queueing systems with feedback of output customers, the time spent by a customer in the system from arrival to final departure (the *sojourn time*) is of a primary interest. The first and second moments of the sojourn time in a first-come first-served M/G/1 queue with Bernoulli feedback were given by Takács [1962]. By using the symbolic formula manipulation functions of *Mathematica*, we can calculate higher-order moments of the sojourn time for Takács's model. We show a *Mathematica* program for this calculation as well as the explicit result for the third moment. The moments of the sojourn time can also be obtained for the exhaustive and gated service systems with multiple server vacations. An attempt shown in this paper suggests a new technique for symbolic computational queueing theory.

### 1. Introduction

In queueing systems with Bernoulli feedback of output customers, each customer whose service is completed instantaneously joins the tail of the queue with probability  $1 - \nu$ , or leaves the system with probability  $\nu$ , where  $0 < \nu \leq 1$ . Such systems can be used to model a number of practical systems in which services may be repeated for some reason. Two noteworthy application examples in communication networks are the packet transmission in an error-prone channel or in a contention-based multiple access channel (where  $\nu$  is the probability of successful transmission), and the segmented message transmission (where the number of segments in a message is geometrically distributed with mean  $1/\nu$ ).

Takács [14] first studies a first-come first-served (FCFS) M/G/1 system with Bernoulli feedback, and derived the expressions for the mean and the second moment of the sojourn time  $T$  of an arbitrary customer. The sojourn time is the time from the arrival to the final departure of a customer, which is one of major performance measures of the system. Further studies on the queue length, the sojourn time, and the waiting time are provided by Disney, McNickle, and Simon [8], Disney [6], and Disney, König, and Schmidt [7]. Takács's model has been extended in several ways. Among them are systems with multiple classes of customers. D'avignon and Disney [4], Simon [13], and Fontana and Berzosa [9, 10] consider priority queues. Takagi [15], Boxma [3], de Moraes [5], Sidi, Levy, and Fuhrmann [12], and Takine, Takagi, and Hasegawa [19] investigate cyclic-service queues (polling systems).

The purpose of this paper is to obtain higher-order moments of the customer sojourn time based on Takács's basic model. Since manual calculation is practically impossible due to the complexity in symbolic manipulation of equations, we capitalize on the computer software *Mathematica* for this calculation [20]. A *Mathematica* program and the explicit expression for the third moment of the sojourn time are shown. We also provide *Mathematica* programs that calculate the moments of the sojourn time in exhaustive and gated service systems with multiple server vacations. These systems are analyzed by Takine, Takagi, and Hasegawa

[19], where only the mean sojourn time are given by manual calculation. We now get higher-order moments of the sojourn time. The higher-order moments can be used to characterize the shape of distribution, such as the coefficients of variation, skewness, and kurtosis.

In a previous work [17, 18], we provided *Mathematica* programs that calculate the moments of the waiting time, sojourn time, queue size, and busy period length, and showed the results for up to the 10th order for an M/G/1 queue without feedback. This method is in contrast with the numerical evaluation of higher-order moments for the length of a busy period by Klimko and Neuts [11]. Our symbolic moment calculation also complements the use of *Mathematica* for the numerical evaluation of formulas and the discrete event simulation for computer performance analysis shown by Allen [1, 2]. All the programs in this paper are written and executed using *Mathematica* Version 2.2 for SPARC from Wolfram Research, Inc. [20].

## 2. Notation and Previous Results

Let us introduce the notation for M/G/1 queueing systems with Bernoulli feedback studied in this paper. The customer arrival rate is denoted by  $\lambda$ . The Laplace-Stieltjes transform (LST) of the distribution function (DF), the mean, and the  $n$ th moment of the service time are denoted by  $B^*(s)$ ,  $b$ , and  $b^{(n)}$  ( $n = 2, 3, \dots$ ), respectively. The Bernoulli feedback mechanism with parameter  $\nu$  is described at the beginning of Section 1. The traffic intensity is then given by  $\lambda b/\nu$ , which is also the server utilization. We assume that  $\lambda b < \nu$  for the stability of the system. The LST of the DF for the sojourn time  $T$  of a customer is denoted by  $T^*(s)$ . For systems with multiple server vacations (the vacationing mechanism is described in Section 4), we denote the LST of the DF for the length  $V$  of each vacation by  $V^*(s)$ .

The results for the moments of the sojourn time  $T$  that were previously available are summarized as follows. For a system without server vacation, we have [14]

$$E[T] = \frac{\lambda b^{(2)} + 2(1 - \lambda b)b}{2(\nu - \lambda b)}, \quad (2.1)$$

$$\begin{aligned} E[T^2] &= \frac{\nu^2 - 2\nu}{6(\nu - \lambda b)^2[\nu^2 - \nu(2 + \lambda b) + \lambda b]} \\ &\times \left\{ 2\nu[6\lambda b^3 - 6b^2 - 6\lambda b b^{(2)} + 3b^{(2)} + \lambda b^{(3)}] \right. \\ &\quad \left. - 12\lambda b^3 + 12b^2 + 6\lambda b b^{(2)} - 2\lambda^2 b b^{(3)} + 3[\lambda b^{(2)}]^2 \right\}. \end{aligned} \quad (2.2)$$

For an exhaustive service system with multiple server vacations, we have [19]

$$E[T] = \frac{\lambda b^{(2)}}{2(\nu - \lambda b)} + \frac{(1 - \lambda b)b}{\nu - \lambda b} + \frac{E[V^2]}{2E[V]}. \quad (2.3)$$

For a gated service system with multiple server vacations, we have [19]

$$E[T] = \frac{\lambda b^{(2)}}{2(\nu - \lambda b)} + \frac{(1 - \lambda b)b + (1 - \nu + \lambda b)E[V]}{\nu - \lambda b} + \frac{E[V^2]}{2E[V]}. \quad (2.4)$$

In the present paper, the higher-order moments for  $T$  are given for these systems.

### 3. M/G/1 System without Server Vacations

We first consider an M/G/1 system without server vacations. According to Takács [14] (see also Takagi [16, Section 1.6]), the LST  $T^*(s)$  of the DF for the sojourn time  $T$  of an arbitrary customer is given by

$$T^*(s) = F^*(1, s), \quad (3.1)$$

where  $F^*(z, s)$  satisfies the functional equation

$$F^*(z, s) = \nu F_1^*(z, s) + (1 - \nu)B^*(s + \lambda - \lambda z)F^*([\nu + (1 - \nu)z]B^*(s + \lambda - \lambda z), s). \quad (3.2)$$

Furthermore,  $F_1^*(z, s)$  is explicitly given by

$$F_1^*(z, s) = (1 - \lambda b/\nu)B^*(s + \lambda - \lambda z) + \Pi^*([\nu + (1 - \nu)z]B^*(s + \lambda - \lambda z), s + \lambda - \lambda z), \quad (3.3)$$

where

$$\begin{aligned} \Pi^*(z, s) &:= \sum_{k=1}^{\infty} z^k \int_0^{\infty} e^{-sx} P\{L = k, x < X_+ \leq x + dx\} \\ &= \frac{\lambda(1 - \lambda b/\nu)z(1 - z)[B^*(\lambda - \lambda z) - B^*(s)]}{(s - \lambda + \lambda z)\{\nu + (1 - \nu)z\}B^*(\lambda - \lambda z) - z}. \end{aligned} \quad (3.4)$$

is the double transform of the joint distribution for the number  $L$  of customers in the system and the remaining service time  $X_+$  at an arbitrary time (= at an arrival time) during a busy period.

Therefore it is in principle straightforward to calculate the  $n$ th moment

$$E[T^n] = (-1)^n \left[ \frac{d^n T^*(s)}{ds^n} \right]_{s=0} = (-1)^n \left[ \frac{\partial^n F^*(z, s)}{\partial s^n} \right]_{z=1, s=0} \quad n = 1, 2, \dots \quad (3.5)$$

of the sojourn time  $T$  from (3.2)–(3.4). The procedure is suggested by W. S. Brown in the Appendix of Takács [14]. For  $n = 1$ , by differentiating both sides of (3.2) with respect to  $z$  and then setting  $z = 1$  and  $s = 0$ , we get

$$\left[ \frac{\partial F^*(z, s)}{\partial z} \right]_{z=1, s=0} = \frac{1}{1 - (1 - \nu)(1 - \nu + \lambda b)} \left\{ (1 - \nu)\lambda b + \nu \left[ \frac{\partial F_1^*(z, s)}{\partial z} \right]_{z=1, s=0} \right\}. \quad (3.6)$$

By differentiating both sides of (3.2) with respect to  $s$  and then setting  $z = 1$  and  $s = 0$ , we get

$$\nu \left[ \frac{\partial F^*(z, s)}{\partial s} \right]_{z=1, s=0} = \nu \left[ \frac{\partial F_1^*(z, s)}{\partial s} \right]_{z=1, s=0} - (1 - \nu)b \left\{ 1 + \left[ \frac{\partial F^*(z, s)}{\partial z} \right]_{z=1, s=0} \right\}. \quad (3.7)$$

Substituting (3.6) into (3.7), we obtain

$$\begin{aligned} E[T] &= - \left[ \frac{\partial F^*(z, s)}{\partial s} \right]_{z=1, s=0} \\ &= \frac{(1 - \nu)(2 - \nu)b}{1 - (1 - \nu)(1 - \nu + \lambda b)} + \frac{(1 - \nu)b}{1 - (1 - \nu)(1 - \nu + \lambda b)} \left[ \frac{\partial F_1^*(z, s)}{\partial z} \right]_{z=1, s=0} \\ &\quad - \left[ \frac{\partial F_1^*(z, s)}{\partial s} \right]_{z=1, s=0}, \end{aligned} \quad (3.8)$$

where  $[\partial F_1^*(z, s)/\partial z]_{z=1, s=0}$  and  $[\partial F_1^*(z, s)/\partial s]_{z=1, s=0}$  can be calculated from (3.3) and (3.4). The result is given in (2.1). This result agrees with the evaluation of the mean sojourn time from Little's theorem and the mean queue size in the corresponding batch-arrival M/G/1 queue in which the batch size is geometrically distributed with mean  $1/\nu$ .

For  $n = 2$ , the differentiation of both sides of (3.2) and evaluation of the derivatives at  $z = 1$  and  $s = 0$  yields three equations with respect to

$$\left[ \frac{\partial^2 F^*(z, s)}{\partial z^2} \right]_{z=1, s=0}, \quad \left[ \frac{\partial^2 F^*(z, s)}{\partial z \partial s} \right]_{z=1, s=0}, \quad \text{and} \quad \left[ \frac{\partial^2 F^*(z, s)}{\partial s^2} \right]_{z=1, s=0}. \quad (3.9)$$

Solving these equations gives

$$E[T^2] = \left[ \frac{\partial^2 F^*(z, s)}{\partial s^2} \right]_{z=1, s=0} \quad (3.10)$$

in terms of five partial derivatives  $[\partial^2 F_1^*(z, s)/\partial z^2]_{z=1, s=0}$ ,  $[\partial^2 F_1^*(z, s)/\partial z \partial s]_{z=1, s=0}$ ,  $[\partial^2 F_1^*(z, s)/\partial s^2]_{z=1, s=0}$ ,  $[\partial F_1^*(z, s)/\partial z]_{z=1, s=0}$ , and  $[\partial F_1^*(z, s)/\partial s]_{z=1, s=0}$ , which are all known from (3.3) and (3.4). We can continue this procedure recursively with respect to  $n$ . However, the complexity in symbolic calculation grows rapidly as  $n$  increases. Brown spends six pages of Takács [14] in order to explain the procedure to calculate  $E[T^2]$ .

We present a *Mathematica* program that calculates  $E[T^n]$  automatically in Figure 1. The first and second moment,  $E[T]$  and  $E[T^2]$ , obtained by this program agree with (2.1) and (2.2), respectively. The third moment  $E[T^3]$  is given in the Appendix.

#### 4. M/G/1 Systems with Multiple Server Vacations

We proceed to study exhaustive and gated service M/G/1 systems with multiple server vacations. In an exhaustive service system with multiple server vacations, the server begins a vacation each time the system becomes empty. If the server returns from a vacation to find the system not empty, it starts to work immediately and continues until the system becomes empty again (exhaustive service). If the server returns from a vacation to find no customers waiting, it begins another vacation immediately, and repeats vacations until it finds at least one customer waiting upon returning from a vacation (multiple vacations). The lengths of successive vacations are independent and identically distributed, and also are independent of the arrival and service processes. In a gated service system, when the server returns from a vacation it accepts and serves continuously only those customers that are waiting at that time, deferring the service to all the messages that arrive during the service period until after the next vacation.

We first consider an exhaustive service system. Takine, Takagi, and Hasegawa [19] show that the LST  $T^*(s)$  of the DF for the sojourn time  $T$  of an arbitrary customer in this system is again given by (3.1) and (3.2). However,  $F_1^*(z, s)$  is now given by

$$\begin{aligned} F_1^*(z, s) &= B^*(s + \lambda - \lambda z) \Omega^*([\nu + (1 - \nu)z]B^*(s + \lambda - \lambda z), s + \lambda - \lambda z) \\ &\quad + \Pi^*([\nu + (1 - \nu)z]B^*(s + \lambda - \lambda z), s + \lambda - \lambda z), \end{aligned} \quad (4.1)$$

where

$$\Omega^*(z, s) = \frac{(\nu - \lambda b)[V^*(\lambda - \lambda z) - V^*(s)]}{\nu E[V](s - \lambda - \lambda z)} \quad (4.2)$$

and

$$\Pi^*(z, s) = \frac{(\nu - \lambda b)z[B^*(\lambda - \lambda z) - B^*(s)][1 - V^*(\lambda - \lambda z)]}{\nu E[V](s - \lambda + \lambda z)\{[\nu + (1 - \nu)z]B^*(\lambda - \lambda z) - z\}}. \quad (4.3)$$

---

```

B[0] = 1 ; t[0] = 1 ; B'[0] = -b
Derivative[n_][B][0] = Derivative[n][b]*(-1)^n
T[s] = F[1,s] ; F[1,0] = 1
F1[z,s] = (1- lambda b /nu) B[s+ lambda - lambda z] +
  Pai[(nu + (1-nu) z) B [s+ lambda - lambda z] , s+lambda - lambda z]
taylor = Series[F1[z,s],{z,1,3},{s,0,3}]
Pai[z,s]:=( ( lambda ( 1 - lambda b / nu ) z ( 1 - z )
  ( B [ lambda - lambda z ] - B [s] ) ) / ( ( s - lambda
  + lambda z ) ( ( nu + ( 1 - nu ) z ) B [ lambda - lambda z ] - z ) ) )
taylorPi = Series[Pai[z,s],{z,1,3},{s,0,3}]
Derivative[n_,m_][Pai][1,0] := Simplify[Coefficient[Coefficient[taylorPi,s,m],
  (-1+z),n]*n!*m!/. {z->1,s->0}]
Derivative[n_,m_][F1][1,0] := Simplify[Coefficient[Coefficient[taylor,s,m],
  (-1+z),n]*n!*m! /. {z->1,s->0}]
(* the functional equation to be solved *)
eq = F[z,s] == nu F1[z,s] + (1-nu) B[s+ lambda - lambda z] *
  F[(nu + (1-nu) z) B[s+ lambda - lambda z],s]
(* the main part *)
getanswers[0] = {}
equations[n_] := Table[D[D[eq,{s,n-i}],{z,i}] /.{z->1,s->0},{i,0,n}]
answers[n_] := Table[Derivative[n-i,i][F][1,0],{i,0,n}]
getanswers[n_] :=getanswers[n] = Union[Solve[equations[n],answers[n]][[1]]
  /. getanswers[n-1],getanswers[n-1]]
Mom[n_] := Mom[n] = Simplify[Derivative[0,n][F][1,0] /. getanswers[n] ]

```

---

Figure 1. Symbolic calculation of the moments of the sojourn time for an FCFS M/G/1 system with Bernoulli feedback without server vacations.

---

A *Mathematica* program that calculates  $E[T^n]$  for an exhaustive service system with multiple server vacation is given in Figure 2. The mean sojourn time  $E[T]$  obtained by this program agrees with (2.3). The second moment is given as follows:

$$E[T^2] = \frac{(2 - \nu)\nu\{E[V]A + E[V^2]B + E[V^3]C\}}{6(\nu - b\lambda)^2(2\nu - \nu^2 - b\lambda + \nu b\lambda)E[V]}, \quad (4.4)$$

where

$$\begin{aligned} A &\equiv 12(-1 + \nu)(-1 + b\lambda)b^2 + 6(\nu + b\lambda - 2\nu b\lambda)b^{(2)} + 3\lambda^2 b^{(2)^2} + 2\lambda(\nu - b\lambda)b^{(3)}, \\ B &\equiv 3(\nu - b\lambda)(2b - 2b^2\lambda + \lambda b^{(2)}), \\ C &\equiv 2(-\nu + b\lambda)^2. \end{aligned} \quad (4.5)$$

For a gated service system, Takine, Takagi, and Hasegawa [19] show that the LST  $T^*(s)$  of the DF for the sojourn time  $T$  of an arbitrary customer is given by (3.1), where  $F^*(z, s)$  now satisfies the functional equation

$$F^*(z, s) = \nu F_1^*(z, s) + (1 - \nu)B^*(s + \lambda - \lambda z)V^*(s + \lambda - \lambda z)F^*([\nu + (1 - \nu)z]B^*(s + \lambda - \lambda z), s) \quad (4.6)$$

---

```

B[0] = 1 ; V[0] = 1 ; t[0] = 1 ; V'[0] = -E[V]
Derivative[n_][V][0] = E[V^n]*(-1)^n
Derivative[n_][B][0] = Derivative[n][b]*(-1)^n
T[s] = F[1,s] ; F[1,0] = 1
F1[z,s] = B[s+ lambda - lambda z] *
  Omega[(nu+ (1-nu) z) B[s + lambda - lambda z], s+lambda - lambda z] +
  Pai[(nu+ (1-nu) z) B [s+ lambda - lambda z] , s+lambda - lambda z]
taylor = Series[F1[z,s],{z,1,3},{s,0,3}]
Pai[z,s] = (nu - lambda b)/(nu E[V]) *
  (z ( B[s] - B[lambda - lambda z])(1 - V[lambda - lambda z]))/
  ((z - (nu + (1-nu)z)B[lambda - lambda z])(s- lambda + lambda z))
taylorPai = Series[Pai[z,s],{z,1,3},{s,0,3}]
Omega[z,s] = (nu - lambda b) / (nu E[V]) *
  (V[lambda - lambda z] - V[s])/(s - lambda + lambda z)
taylorOmega = Series[Omega[z,s],{z,1,3},{s,0,3}]
Derivative[n_,m_][Pai][1,0] := Simplify[Coefficient[Coefficient[taylorPai,s,m],
  (-1+z),n]*n!*m!/. {z->1,s->0}]
Derivative[n_,m_][Omega][1,0] :=
  Simplify[Coefficient[Coefficient[taylorOmega,s,m],(-1+z),n] * n! * m!
  /. {z->1,s->0}]
Derivative[n_,m_][F1][1,0] := Simplify[Coefficient[Coefficient[taylor,s,m],
  (-1+z),n]*n!*m! /. {z->1,s->0}]
Pai[1,0] = Coefficient[Coefficient[taylorPai,s,0],(-1+z),0]
Omega[1,0] = Coefficient[Coefficient[taylorOmega,s,0],(-1+z),0]
(* the functional equation to be solved *)
eq = F[z,s] == nu F1[z,s] + (1-nu) B[s+ lambda - lambda z] *
  F[(nu + (1-nu) z) B[s + lambda - lambda z],s]
(* the main part *)
getanswers[0] = {}
equations[n_] := Table[D[D[eq,{s,n-i}],{z,i}] /. {z->1,s->0},{i,0,n}]
answers[n_] := Table[Derivative[n-i,i][F][1,0],{i,0,n}]
getanswers[n_] := getanswers[n] = Union[Solve[equations[n],answers[n]][[1]]
  /. getanswers[n-1], getanswers[n-1]]
Mom[n_] := Mom[n] = Simplify[Derivative[0,n][F][1,0] /. getanswers[n]]

```

---

Figure 2. Symbolic calculation of the moments of the sojourn time for an exhaustive FCFS M/G/1 system with Bernoulli feedback with multiple server vacations.

---

```

B[0] = 1 ; V[0] = 1 ; t[0] = 1 ; Q[1] = 1 ; B'[0] = -b ; V'[0] = -E[V]
Derivative[n_][V][0] = E[V^n]*(-1)^n
Derivative[n_][B][0] = Derivative[n][b]*(-1)^n
T[s] = F[1,s] ; F[1,0] = 1
F1[z,s] = B[s+ lambda - lambda z] *
  Omega[(nu+ (1-nu) z) B[s + lambda - lambda z], s+lambda - lambda z] +
  Pai[(nu+ (1-nu) z) B [s+ lambda - lambda z] , s+lambda - lambda z] *
  V[s + lambda - lambda z]
taylor = Series[F1[z,s],{z,1,4},{s,0,4}]
Pai[z,s] = (nu - lambda b)/( nu E[V]) *
  (z ( B[s] - B[lambda - lambda z]))/( s - lambda + lambda z) *
  (Q[(nu + (1-nu) z ) B[lambda - lambda z]]-Q[z])/
  (z - ( nu + (1-nu) z) B[lambda - lambda z])
taylorPi = Series[Pai[z,s],{z,1,4},{s,0,4}]
Omega[z,s] = (nu - lambda b) / (nu E[V]) *
  (V[lambda - lambda z ] - V[s])/(s - lambda + lambda z) *
  Q[(nu + (1-nu)z)B[lambda - lambda z]]
taylorOmega = Series[Omega[z,s],{z,1,3},{s,0,3}]
eqq = Q[z] == Q[(nu + (1-nu) z) B[lambda - lambda z]] V[lambda - lambda z]
q1 = Solve[ D[eqq,z] /. z->1 , Derivative[1][Q][1]]
Derivative[1][Q][1] = Derivative[1][Q][1] /. q1 [[1]]
q2 = Solve[ D[eqq,{z,2}] /. z->1 , Derivative[2][Q][1]]
Derivative[2][Q][1] = Derivative[2][Q][1] /. q2 [[1]]
q3 = Solve[ D[eqq,{z,3}] /. z->1 , Derivative[3][Q][1]]
Derivative[3][Q][1] = Derivative[3][Q][1] /. q3 [[1]]
Derivative[n_,m_][Pai][1,0] := Simplify[Coefficient[Coefficient[taylorPi,s,m],
  (-1+z),n]*n!*m!/. {z->1,s->0}]
Derivative[n_,m_][Omega][1,0] :=
Simplify[Coefficient[Coefficient[taylorOmega,s,m],(-1+z),n] * n! * m!
  /. {z->1,s->0} ]
Derivative[n_,m_][F1][1,0] := Simplify[Coefficient[Coefficient[taylor,s,m],
  (-1+z),n]*n!*m! /. {z->1,s->0}]
Pai[1,0] = Coefficient[Coefficient[taylorPi,s,0],(-1+z),0]
Omega[1,0] = Coefficient[Coefficient[taylorOmega,s,0],(-1+z),0]
(* the functional equation to be solved *)
eq = F[z,s] == nu F1[z,s] + (1-nu) B[s+ lambda - lambda z] *
  F[(nu + (1-nu) z) B[s + lambda - lambda z],s] * V[s + lambda - lambda z]
(* the main part *)
getanswers[0] = {}
equations[n_] := Table[D[D[eq,{s,n-i}],{z,i}] /.{z->1,s->0},{i,0,n}]
answers[n_] := Table[Derivative[n-i,i][F][1,0],{i,0,n}]
getanswers[n_] := getanswers[n] = Union[Solve[equations[n],answers[n]]][[1]]
  /. getanswers[n-1],getanswers[n-1]]
Mom[n_] := Mom[n] = Simplify[Derivative[0,n][F][1,0] /. getanswers[n] ]

```

---

Figure 3. Symbolic calculation of the moments of the sojourn time for a gated FCFS M/G/1 system with Bernoulli feedback with multiple server vacations.

---

and

$$F_1^*(z, s) = B^*(s + \lambda - \lambda z)\Omega^*([\nu + (1 - \nu)z]B^*(s + \lambda - \lambda z), s + \lambda - \lambda z) \\ + V^*(s + \lambda - \lambda z)\Pi^*([\nu + (1 - \nu)z]B^*(s + \lambda - \lambda z), s + \lambda - \lambda z) \quad (4.7)$$

with

$$\Omega^*(z, s) = \frac{(\nu - \lambda b)Q([\nu + (1 - \nu)z]B^*(\lambda - \lambda z))[V^*(\lambda - \lambda z) - V^*(s)]}{\nu E[V](s - \lambda + \lambda z)}, \quad (4.8)$$

$$\Pi^*(z, s) = \frac{(\nu - \lambda b)z\{Q([\nu + (1 - \nu)z]B^*(\lambda - \lambda z)) - Q(z)\}[B^*(\lambda - \lambda z) - B^*(s)]}{\nu E[V](s - \lambda + \lambda z)\{[\nu + (1 - \nu)z]B^*(\lambda - \lambda z) - z\}}, \quad (4.9)$$

and

$$Q(z) = Q([\nu + (1 - \nu)z]B^*(\lambda - \lambda z))V^*(\lambda - \lambda z). \quad (4.10)$$

A *Mathematica* program that calculates  $E[T^n]$  for a gated service system with multiple server vacation is given in Figure 3. The mean sojourn time  $E[T]$  obtained by this program agrees with (2.4). The second moment is given as follows:

$$E[T^2] = \frac{(-2 + \nu)\{E[V]A + E[V]^2B + E[V]^3C + E[V^2]D + E[V^2]E[V]F + E[V^3]G\}}{6(-2 - b\lambda + \nu)(-b\lambda + \nu)^2(b\lambda - 2\nu - b\lambda\nu + \nu^2)E[V]}, \quad (4.11)$$

where

$$A \equiv \nu(-2 - b\lambda + \nu)(12b^2(-1 + b\lambda)(-1 + \nu) + 6(b\lambda + \nu - 2b\lambda\nu)b^{(2)} \\ + 3\lambda^2 b^{(2)^2} + 2\lambda(-b\lambda + \nu)b^{(3)}), \\ B \equiv 12b(b\lambda - 4\nu - 3b\lambda\nu + 2b^2\lambda^2\nu + b^3\lambda^3\nu + 6\nu^2 - 2b^2\lambda^2\nu^2 - 2\nu^3 + b\lambda\nu^3) \\ + 6\lambda(b\lambda - 4\nu - 4b\lambda\nu - b^2\lambda^2\nu + 4\nu^2 + 2b\lambda\nu^2 - \nu^3)b^{(2)}, \\ C \equiv 12(1 + b\lambda - \nu)(b\lambda - 2\nu - 3b\lambda\nu - b^2\lambda^2\nu + 3\nu^2 + 2b\lambda\nu^2 - \nu^3), \\ D \equiv 3(b\lambda - \nu)\nu(-2 - b\lambda + \nu)(-2b + 2b^2\lambda - \lambda b^{(2)}), \\ F \equiv 6(b\lambda - \nu)(-b\lambda + 4\nu + 6b\lambda\nu + 2b^2\lambda^2\nu - 6\nu^2 - 4b\lambda\nu^2 + 2\nu^3), \\ G \equiv 2\nu(-2 - b\lambda + \nu)(-b\lambda + \nu)^2. \quad (4.12)$$

## 5. Concluding Remarks

We have shown *Mathematica* programs for the symbolic calculation of the moments for the sojourn time in M/G/1 queues with Bernoulli feedback. These programs only simulate manual calculation, which is rather straightforward. Much efforts are needed after the calculation in order to present the results in a form with simple appearance. It will be of interest to explore the capability of *Mathematica* for the study of queues and other performance evaluation models.

## References

- [1] A. O. Allen: *Probability, Statistics, and Queueing Theory with Computer Science Applications*, Second edition (Academic Press, San Diego, California, 1990).
- [2] A. O. Allen: *Introduction to Computer Performance Analysis with Mathematica* (Academic Press, Boston, 1994).
- [3] O. J. Boxma: Workloads and waiting times in single-server systems with multiple customer classes. *Queueing Systems*, **5** (1989) 185–214.



- [4] G. R. D'avignon and R. L. Disney: Queues with instantaneous feedback. *Management Science*, **24** (1977) 168–180.
- [5] L. F. M. de Moraes: Comments on “Analysis and applications of a multiqueue cyclic service system with feedback.” *IEEE Transactions on Communications*, **38** (1980) 148–149.
- [6] R. L. Disney: A note on sojourn times in M/G/1 queues with instantaneous Bernoulli feedback. *Naval Research Logistics Quarterly*, **28** (1981) 679–684.
- [7] R. L. Disney, D. König, and V. Schmidt: Stationary queue-length and waiting-time distributions in single-server feedback queues. *Advances in Applied Probability*, **16** (1984) 437–446.
- [8] R. L. Disney, D. C. McNickle, and B. Simon: The M/G/1 queue with instantaneous Bernoulli feedback. *Naval Research Logistics Quarterly*, **27** (1980) 635–644.
- [9] B. Fontana and C. D. Berzosa: Stationary queue-length distribution in an M/G/1 queue with two non-preemptive priorities and general feedback. In W. Bux and H. Rudin (eds.): *Performance of Computer-Communication Systems* (Elsevier, Amsterdam, 1984), 333–347.
- [10] B. Fontana and C. D. Berzosa: M/G/1 queue with N-priorities and feedback: joint queue-length distributions and response time distribution for any particular sequence. In M. Akiyama (ed.): *Teletraffic Issues in an Advanced Information Society, ITC-11* (Elsevier, Amsterdam, 1985), 452–458.
- [11] E. Klimko and M. F. Neuts: The single server queue in discrete time – numerical analysis II. *Naval Research Logistics Quarterly*, **20** (1973) 305–319.
- [12] M. Sidi, H. Levy, and S. W. Fuhrmann: A queueing network with a single cyclically roving server. *Queueing Systems*, **11** (1992) 121–144. A corrigendum in “Correction to equation (5.6) in the paper: A queueing network with a single cyclically roving server.” *Queueing Systems*, **16** (1994) 193.
- [13] B. Simon: Priority queues with feedback. *Journal of Association for Computing Machinery*, **31** (1984) 134–149.
- [14] L. Takács: A single-server queue with feedback. *The Bell System Technical Journal*, **42** (1963) 505–519.
- [15] H. Takagi: Analysis and applications of a multi-queue cyclic service system with feedback. *IEEE Transactions on Communications*, **COM-35** (1987) 248–250.
- [16] H. Takagi: *Queueing Analysis : A Foundation of Performance Evaluation, Vol. 1: Vacation and Priority Systems, Part 1* (Elsevier, Amsterdam, 1991).
- [17] H. Takagi and K. Sakamaki: Symbolic moment calculation for an M/G/1 queue. Discussion Paper No. 596, Institute of Socio-Economic Planning, University of Tsukuba, September 1994.
- [18] H. Takagi and K. Sakamaki: Moments for M/G/1 queues. *The Mathematica Journal*, **6** (1996), 75–80.
- [19] T. Takine, H. Takagi, and T. Hasegawa: Sojourn times in vacation and polling systems with Bernoulli feedback. *Journal of Applied Probability*, **28** (1991) 422–432.
- [20] S. Wolfram: *Mathematica: A System for Doing Mathematics by Computer*, Second edition (Addison-Wesley, Reading, Massachusetts, 1991).

### Appendix: Third Moment of the Sojourn Time in an FCFS M/G/1 System with Bernoulli Feedback

$$E[T^3] = \frac{B}{A}$$

where

$$A \equiv 4(b\lambda - \nu)^3(-b\lambda + 2\nu + b\lambda\nu - \nu^2)^2 \\ \times (-2b\lambda - b^2\lambda^2 + 3\nu + 4b\lambda\nu + b^2\lambda^2\nu - 3\nu^2 - 2b\lambda\nu^2 + \nu^3),$$

and

$$B \equiv -96b^5\lambda^2\nu + 48b^6\lambda^3\nu + 48b^7\lambda^4\nu - 48b^4\lambda\nu^2 + 432b^5\lambda^2\nu^2 - 168b^6\lambda^3\nu^2 - 216b^7\lambda^4\nu^2 + \\ 288b^3\nu^3 + 96b^4\lambda\nu^3 - 1056b^5\lambda^2\nu^3 + 288b^6\lambda^3\nu^3 + 384b^7\lambda^4\nu^3 - 1152b^3\nu^4 + 168b^4\lambda\nu^4 + \\ 1656b^5\lambda^2\nu^4 - 336b^6\lambda^3\nu^4 - 336b^7\lambda^4\nu^4 + 1896b^3\nu^5 - 720b^4\lambda\nu^5 - 1584b^5\lambda^2\nu^5 + 264b^6\lambda^3\nu^5 + \\ 144b^7\lambda^4\nu^5 - 1656b^3\nu^6 + 936b^4\lambda\nu^6 + 864b^5\lambda^2\nu^6 - 120b^6\lambda^3\nu^6 - 24b^7\lambda^4\nu^6 + 816b^3\nu^7 - \\ 600b^4\lambda\nu^7 - 240b^5\lambda^2\nu^7 + 24b^6\lambda^3\nu^7 - 216b^3\nu^8 + 192b^4\lambda\nu^8 + 24b^5\lambda^2\nu^8 + 24b^6\lambda^3\nu^8 - 24b^7\lambda^4\nu^8 \\ + (-48b^4\lambda^3\nu - 24b^5\lambda^4\nu - 120b^3\lambda^2\nu^2 + 228b^4\lambda^3\nu^2 + 168b^5\lambda^4\nu^2 + 96b^2\lambda\nu^3 + 576b^3\lambda^2\nu^3 - \\ 468b^4\lambda^3\nu^3 - 396b^5\lambda^4\nu^3 + 288b\nu^4 - 564b^2\lambda\nu^4 - 1284b^3\lambda^2\nu^4 + 576b^4\lambda^3\nu^4 + 420b^5\lambda^4\nu^4 - \\ 864b\nu^5 + 1236b^2\lambda\nu^5 + 1536b^3\lambda^2\nu^5 - 444b^4\lambda^3\nu^5 - 204b^5\lambda^4\nu^5 + 1032b\nu^6 - 1380b^2\lambda\nu^6 - \\ 984b^3\lambda^2\nu^6 + 192b^4\lambda^3\nu^6 + 36b^5\lambda^4\nu^6 - 624b\nu^7 + 852b^2\lambda\nu^7 + 312b^3\lambda^2\nu^7 - 36b^4\lambda^3\nu^7 + \\ 192b\nu^8 - 276b^2\lambda\nu^8 - 36b^3\lambda^2\nu^8 - 24b\nu^9 + 36b^2\lambda\nu^9) b^{(2)} \\ + (-30b^3\lambda^4\nu - 6b^4\lambda^5\nu - 12b^2\lambda^3\nu^2 + 42b^3\lambda^4\nu^2 + 30b^4\lambda^5\nu^2 - 30b\lambda^2\nu^3 + 192b^2\lambda^3\nu^3 + \\ 36b^3\lambda^4\nu^3 - 48b^4\lambda^5\nu^3 + 180\lambda\nu^4 + 6b\lambda^2\nu^4 - 474b^2\lambda^3\nu^4 - 102b^3\lambda^4\nu^4 + 30b^4\lambda^5\nu^4 - 450\lambda\nu^5 + \\ 78b\lambda^2\nu^5 + 474b^2\lambda^3\nu^5 + 66b^3\lambda^4\nu^5 - 6b^4\lambda^5\nu^5 + 456\lambda\nu^6 - 108b\lambda^2\nu^6 - 216b^2\lambda^3\nu^6 - 12b^3\lambda^4\nu^6 - \\ 234\lambda\nu^7 + 60b\lambda^2\nu^7 + 36b^2\lambda^3\nu^7 + 60\lambda\nu^8 - 12b\lambda^2\nu^8 - 6\lambda\nu^9) [b^{(2)}]^2 \\ + (-3b^3\lambda^6 + 12b^2\lambda^5\nu + 6b^3\lambda^6\nu - 45b\lambda^4\nu^2 - 39b^2\lambda^5\nu^2 - 3b^3\lambda^6\nu^2 + 54\lambda^3\nu^3 + 132b\lambda^4\nu^3 + \\ 51b^2\lambda^5\nu^3 - 117\lambda^3\nu^4 - 144b\lambda^4\nu^4 - 30b^2\lambda^5\nu^4 + 99\lambda^3\nu^5 + 69b\lambda^4\nu^5 + 6b^2\lambda^5\nu^5 - 39\lambda^3\nu^6 - \\ 12b\lambda^4\nu^6 + 6\lambda^3\nu^7) [b^{(2)}]^3 \\ + (20b^4\lambda^4\nu + 4b^5\lambda^5\nu - 24b^3\lambda^3\nu^2 - 48b^4\lambda^4\nu^2 - 20b^5\lambda^5\nu^2 - 12b^2\lambda^2\nu^3 - 12b^3\lambda^3\nu^3 + \\ 56b^4\lambda^4\nu^3 + 32b^5\lambda^5\nu^3 - 32b\lambda\nu^4 + 152b^2\lambda^2\nu^4 + 132b^3\lambda^3\nu^4 - 44b^4\lambda^4\nu^4 - 20b^5\lambda^5\nu^4 + \\ 48\nu^5 - 332b^2\lambda^2\nu^5 - 160b^3\lambda^3\nu^5 + 20b^4\lambda^4\nu^5 + 4b^5\lambda^5\nu^5 - 96\nu^6 + 88b\lambda\nu^6 + 304b^2\lambda^2\nu^6 + \\ 76b^3\lambda^3\nu^6 - 4b^4\lambda^4\nu^6 + 76\nu^7 - 104b\lambda\nu^7 - 128b^2\lambda^2\nu^7 - 12b^3\lambda^3\nu^7 - 28\nu^8 + 48b\lambda\nu^8 + \\ 20b^2\lambda^2\nu^8 + 4\nu^9 - 8b\lambda\nu^9) b^{(3)} \\ + (4b^4\lambda^6 - 18b^3\lambda^5\nu - 8b^4\lambda^6\nu + 60b^2\lambda^4\nu^2 + 50b^3\lambda^5\nu^2 + 4b^4\lambda^6\nu^2 - 106b\lambda^3\nu^3 - 178b^2\lambda^4\nu^3 - \\ 56b^3\lambda^5\nu^3 + 60\lambda^2\nu^4 + 262b\lambda^3\nu^4 + 200b^2\lambda^4\nu^4 + 30b^3\lambda^5\nu^4 - 126\lambda^2\nu^5 - 252b\lambda^3\nu^5 - 100b^2\lambda^4\nu^5 - \\ 6b^3\lambda^5\nu^5 + 104\lambda^2\nu^6 + 110b\lambda^3\nu^6 + 18b^2\lambda^4\nu^6 - 40\lambda^2\nu^7 - 18b\lambda^3\nu^7 + 6\lambda^2\nu^8) b^{(2)}b^{(3)} \\ + (-b^5\lambda^6 + 5b^4\lambda^5\nu + 2b^5\lambda^6\nu - 15b^3\lambda^4\nu^2 - 12b^4\lambda^5\nu^2 - b^5\lambda^6\nu^2 + 31b^2\lambda^3\nu^3 + 42b^3\lambda^4\nu^3 + \\ 11b^4\lambda^5\nu^3 - 32b\lambda^2\nu^4 - 80b^2\lambda^3\nu^4 - 45b^3\lambda^4\nu^4 - 5b^4\lambda^5\nu^4 + 12\lambda\nu^5 + 72b\lambda^2\nu^5 + 80b^2\lambda^3\nu^5 + \\ 22b^3\lambda^4\nu^5 + b^4\lambda^5\nu^5 - 24\lambda\nu^6 - 64b\lambda^2\nu^6 - 36b^2\lambda^3\nu^6 - 4b^3\lambda^4\nu^6 + 19\lambda\nu^7 + 26b\lambda^2\nu^7 + \\ 6b^2\lambda^3\nu^7 - 7\lambda\nu^8 - 4b\lambda^2\nu^8 + \lambda\nu^9) b^{(4)}.$$

Hideaki Takagi

Institute of Policy and Planning Sciences

University of Tsukuba

1-1-1 Tennoudai, Tsukuba-shi

Ibaraki 305-8573

Japan

E-mail: [takagi@shako.sk.tsukuba.ac.jp](mailto:takagi@shako.sk.tsukuba.ac.jp)